Véges/végtelen (nemtriviális) minták felismerése, limeszek megsejtése

■ Relatív prímek gyakorisága

Kérdés: Milyen gyakran fordulnak elô relatív prímek, azaz mi a valószínûsége, hogy ha valaki véletlenszerûen választ két természetes számot, akkor azok relatív prímek lesznek?

A megoldáshoz a Mathematica built–in függvényét is használhatjuk.

```
?CoprimeQ
```

CoprimeQ[n_1, n_2] yields True if n_1 and n_2 are relatively prime, and yields False otherwise.
CoprimeQ[n_1, n_2, …] yields True if all pairs of the n_i are relatively prime, and yields False otherwise. ≫

```
EvenQ[13]
False
EvenQ[10]
True
OddQ[11]
True
PrimeQ[34]
False
```

Mintahívások

```
CoprimeQ[2, 4]
False
CoprimeQ[2, 5]
True
M = 10;
Table[n^3, {n, 1, M}]
{1, 8, 27, 64, 125, 216, 343, 512, 729, 1000}
```

```
TableForm[Table[{j, k}, {j, 1, 4}, {k, 1, j}]]
```

```
1
1
2    2
1    2
3    3    3
1    2    3
4    4    4    4
1    2    3    4
```

```
TableForm[Table[Binomial[j, k], {j, 0, 6}, {k, 0, j}]]
```

```
1
1    1
1    2    1
1    3    3    1
1    4    6    4    1
1    5    10   10   5    1
1    6    15   20   15   6    1
```

```
TableForm[Table[{j, k}, {j, 1, M}, {k, 1, M / 2}]]
```

```
1     1     1     1     1
1     2     3     4     5
2     2     2     2     2
1     2     3     4     5
3     3     3     3     3
1     2     3     4     5
4     4     4     4     4
1     2     3     4     5
5     5     5     5     5
1     2     3     4     5
6     6     6     6     6
1     2     3     4     5
7     7     7     7     7
1     2     3     4     5
8     8     8     8     8
1     2     3     4     5
9     9     9     9     9
1     2     3     4     5
10    10    10    10    10
1     2     3     4     5
```

Véges mintát generálunk, vizsgájuk azon síkbeli (n,m) rácspontokat, melyekre: $1 \le m$, $n \le M$

```
Show[Graphics[{Red, PointSize[.02], Point[Flatten[Table[{m, n}, {m, 5}, {n, 5}], 1]]}],
 AxesOrigin → {0, 0}, Axes → True]
```



Teljesen eveszti a struktúrát

```
Flatten[{{1, 2}, {3, 4}}]
```

```
{1, 2, 3, 4}
```

```
TreeForm[{{1, 2}, {3, 4}}]
```



Csak bizonyos szintig történik meg az “egyszerûsítés

```
Flatten[{{1, {2}}, {3, 4}, {5, {6}}}, 1]
```

```
{1, {2}, 3, 4, 5, {6}}
```

1. lehetséges megközelítés:
— A rendezett párról eldöntjük a tulajdonságot (True/False), majd konvertájuk az információt (0/1)-re, majd összegzünk

```
Length[Flatten[Table[{n, m}, {n, M}, {m, M}], 1]]
```

```
100
```

```
x^2 + 2 x + 5 /. x → -1
```

```
4
```

```
(Plus @@ Flatten[Table[CoprimeQ[n, m], {n, M}, {m, M}], 1] /. {True → 1, False → 0}) / M^2
```

$\frac{63}{100}$

```
M = 10^3;
```

```
(Plus @@ Flatten[Table[CoprimeQ[n, m], {n, M}, {m, M}], 1] /. {True → 1, False → 0}) / M^2
```

$\frac{608383}{1000000}$

```
Remove[M];
```

```
Table[M = 10^n;
 (Plus @@ Flatten[Table[CoprimeQ[n, m], {n, M}, {m, M}], 1] /. {True → 1, False → 0}) / M^2,
 {n, 1, 3}]
```

$\left\{\frac{63}{100}, \frac{6087}{10000}, \frac{608383}{1000000}\right\}$

```
N[%]
```

{0.63, 0.6087, 0.608383}

Azt sejthetjük, hogy van limesze a sorozatunknak, talán a határértéket is megsejthetjük ($6/\pi^2$)

```
N[6/π^2, 10]
```

0.6079271019

Kitérô: algebrai számok rekonsrukciója közelítô értékek alapján (RootApproximant)

```
N[Sqrt[2] + 1, 10]
```

2.414213562

```
RootApproximant[2.414213562]
```

$1 + \sqrt{2}$

2. lehetséges megközelítés:
—Szelektáljuk a tulajdonság alapján a “jó” párokat majd megszámoljuk ôket (lsd Cases vagy Select)

```
Length[Cases[Flatten[Table[CoprimeQ[n, m], {n, M}, {m, M}], 1], True]]
```

63

```
Length[Cases[Flatten[Table[CoprimeQ[n, m], {n, M}, {m, M}], 1], True]] / M^2
```

$\frac{63}{100}$

```
Length[Select[Flatten[Table[{n, m}, {n, M}, {m, M}], 1], CoprimeQ[Sequence @@ #] &]] / M^2
```

$\frac{63}{100}$

```
M = 10;
```

```
Length[Select[Flatten[Table[{n, m}, {n, M}, {m, M}], 1], CoprimeQ[#[[1]], #[[2]]] &]] / M^2
```

$\frac{63}{100}$

Kitérô: “Tiszta függvények”

```
f2[x_] := x^2
```

```
f2[5]
```

25

```
(#^2) &[5]
```

25

```
Function[{x}, x^2][6]
```

36

```
{2, 3}
```

```
Map[#^2 - 2&, {3, 7, 2, 13}]
```

{7, 47, 2, 167}

```
f[n_] := If[PrimeQ[n], n + 5, n^2]
```

```
f[5]
```

10

```
Graphics[{PointSize[.02],
  Flatten[Table[{If[CoprimeQ[m, n], Red, Blue], Point[{m, n}]}, {m, M}, {n, M}], 1]},
 AxesOrigin → {0, 0}, Axes → True]
```



Kaptunk egy rel. gyakoriságot. Kísérlet: növeljük M értékét, mit lehet megfigyelni (HF), milyen mintát kapunk?

Kiegészítô magyarázatok a felhasznált Mathematica függvényekhez.

A. Cases (szintaktikai minta)

```
Cases[{1 + I, 5, 7 - I, Sqrt[2], 10, 12}, x_Complex]
```

{1 + ⅈ, 7 - ⅈ}

B. Select (szemantikai minta)

```
Select[{1, 5, 7, 9, 10, 12}, EvenQ[#] &]
```

{10, 12}

```
Select[{1, 5, 7, 9, 10, 12}, EvenQ]
```

{10, 12}

```
Select[Table[n, {n, 100}], PrimeQ]
```

{2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97}

Négyzetmentes számok gyakorisága

Hasonló módszerrel dolgozd fel a négyzetmentes számok gyakoriságára vonatkozó kérdést!

Memo: SQRFreeQ(a)$\Longleftrightarrow$ Ha a kanonikus felbontában legfeljebb elsô hatványon szerepel minden prím.
Hint: SquareFreeQ[a]

```
TableForm[Table[{n, SquareFreeQ[n], FactorInteger[n]}, {n, 15}]]
```

1	True	1 1
2	True	2 1
3	True	3 1
4	False	2 2
5	True	5 1
6	True	2 1 3 1
7	True	7 1
8	False	2 3
9	False	3 2
10	True	2 1 5 1
11	True	11 1
12	False	2 2 3 1
13	True	13 1
14	True	2 1 7 1
15	True	3 1 5 1

```
M = 100;
```

```
Length[Select[Table[n, {n, 1, M}], SquareFreeQ]] / M
```

$\frac{61}{100}$

```
N[%]
```

0.61

```
Remove[M]
```

```
Table[Length[Select[Table[n, {n, 1, M}], SquareFreeQ]] / M, {M, {10^2, 10^3, 10^4, 10^5, 10^6}}]
```

$\left\{\frac{61}{100}, \frac{76}{125}, \frac{6083}{10000}, \frac{30397}{50000}, \frac{303963}{500000}\right\}$

```
N[%, 4]
```

{0.6100, 0.6080, 0.6083, 0.6079, 0.6079}

Újból azt sejthetjük, hogy van limesze a sorozatunknak, talán a határértéket is megsejthetjük $(6/\pi^2)$

```
N[6/π^2]
```

0.607927

```
Length[{5, 6, 1}]
```

3

```
Table[n, {n, 1, M}]
```

{1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21, 22, 23, 24, 25, 26, 27, 28, 29, 30, 31, 32, 33, 34, 35, 36, 37, 38, 39, 40, 41, 42, 43, 44, 45, 46, 47, 48, 49, 50, 51, 52, 53, 54, 55, 56, 57, 58, 59, 60, 61, 62, 63, 64, 65, 66, 67, 68, 69, 70, 71, 72, 73, 74, 75, 76, 77, 78, 79, 80, 81, 82, 83, 84, 85, 86, 87, 88, 89, 90, 91, 92, 93, 94, 95, 96, 97, 98, 99, 100}

```
Select[Table[n, {n, 1, M}], SquareFreeQ]
```

```
{1, 2, 3, 5, 6, 7, 10, 11, 13, 14, 15, 17, 19, 21, 22, 23, 26, 29, 30, 31,
 33, 34, 35, 37, 38, 39, 41, 42, 43, 46, 47, 51, 53, 55, 57, 58, 59, 61, 62, 65,
 66, 67, 69, 70, 71, 73, 74, 77, 78, 79, 82, 83, 85, 86, 87, 89, 91, 93, 94, 95, 97}
```

*Kitérô (abc)

A matematika (számelmélet) egyik fontos (könnyen érthetô) sejtése az abc sejtés. Azt mondja, hogy Ha a+b=c (gcd(a,b)=1) akkor c<sqrtfp(abc) majdnem minden rendezett hármasra.
Vagy ami ezzel ekvialens: q(a,b,c):=log(c)/log(sqrfp(a b c))=1+ϵ csak nagyon kevés (véges sok) számra teljesül.

Nemrég egy japán matematikus ismét bejeletette, hogy megvan a bizonyítás.

http://en.wikipedia.org/wiki/Abc_conjecture

http://index.hu/tudomany/2012/09/12/meglehet_a_primszamok_kozotti_kapcsolat/

Az eddigiekkel szemben most a tulajdonságunkat/predikátumunkat magunk programozzuk le, nincs beépített Mathematica függvény erre.
A függvény egy egész szám négyetmentes részét adja meg, felhasználja a Mathematica faktorizáló függvényét (FactorInteger).
Erre épül rá a q (quality) mennyiség definíciója.

```
SquareFreePart[a_] := Times @@ FactorInteger[a][[All, 1]]
```

```
SquareFreePart[300]
```

```
30
```

```
FactorInteger[300]
```

```
{{2, 2}, {3, 1}, {5, 2}}
```

```
q[{a_, b_}] := N[Log[a + b] / Log[SquareFreePart[a b (a + b)]]]
```

```
q[{4, 127}]
```

```
0.468205
```

```
q[{3, 125}]
```

```
1.42657
```

* Rekonstruáljuk/ellenôrizzük a leideni projekt elsô két sorát ($c < 10^3$) (lsd wiki)!

```
T1 = Flatten[Table[{a, b}, {a, 1, 100}, {b, a + 1, 100}], 1];
```

```
T1[[1]]
```

```
{1, 2}
```

```
Last[T1]
```

```
{99, 100}
```

```
Length[T1]
```

```
4950
```

```
T2 = Select[T1, CoprimeQ[#[[1]], #[[2]]] &];
```

```
Length[T2]
```

3043

```
T3 = Select[T2, #[[1]] + #[[2]] < 100 &];
```

```
Select[T3, q[#] > 1 &]
```

{{1, 8}, {1, 48}, {1, 63}, {1, 80}, {5, 27}, {32, 49}}

```
WS1 = Table[Length[Select[T3, q[#] > qq &]], {qq, {1, 1.05, 1.1, 1.2, 1.3, 1.4}}]
```

{6, 4, 4, 2, 0, 0}

```
T1 = Flatten[Table[{a, b}, {a, 1, 1000}, {b, a + 1, 1000}], 1];
```

```
T1[[1]]
```

{1, 2}

```
Last[T1]
```

{999, 1000}

```
Length[T1]
```

499 500

```
T2 = Select[T1, CoprimeQ[#[[1]], #[[2]]] &];
```

```
Length[T2]
```

304 191

```
T3 = Select[T2, #[[1]] + #[[2]] < 1000 &];
```

```
Select[T3, q[#] > 1 &]
```

{{1, 8}, {1, 48}, {1, 63}, {1, 80}, {1, 224}, {1, 242}, {1, 288}, {1, 512},
{1, 624}, {1, 675}, {1, 728}, {1, 960}, {2, 243}, {3, 125}, {4, 121}, {5, 27},
{5, 507}, {7, 243}, {13, 243}, {25, 704}, {27, 512}, {32, 49}, {32, 343}, {49, 576},
{81, 175}, {81, 544}, {100, 243}, {104, 625}, {169, 343}, {200, 529}, {343, 625}}

```
WS2 = Table[Length[Select[T3, q[#] > qq &]], {qq, {1, 1.05, 1.1, 1.2, 1.3, 1.4}}]
```

{31, 17, 14, 8, 3, 1}

$\texttt{TableForm}\left[\{\texttt{WS1, WS2}\}, \texttt{TableHeadings} \to \left\{\{\texttt{"}10^2\texttt{", "}10^3\texttt{"}\}, \{1, 1.05, 1.1, 1.2, 1.3, 1.4\}\right\}\right]$

	1	1.05	1.1	1.2	1.3	1.4
10^2	6	4	4	2	0	0
10^3	31	17	14	8	3	1